C3M1

Three-Dimensional Graphics

There are several packages of programs in Maple that we will find useful. For many calculus operations we will need "student". For vector operations the package is "linalg". One of the real strengths of Maple is its ability to graph curves and surfaces in a three-dimensional coordinate system. We need the package "plots" in order to do this. There are two basic ways to use the command "plot3d". The first plots $z = f(x, y)$, while the second and most versatile plots $[x, y, z]$ parametrically with x, y, and z as functions of two variables. The reader is encouraged to input the commands being discussed and to try the suggestions to see the effects that they have. It may be to your advantage to save the Maple work that you type in to test because you may be able to cut, paste, and edit them when you need similar entries later. Remember, always start at the top of a worksheet and hit <Enter> all the way down if you have edited the worksheet. The command `restart:` clears the Maple kernel of all internal memory. Some put `restart:` on the first line of a worksheet before any packages such as 'student' or 'plots' so that confusion is avoided if one <Enter>'s from the first line all the way through. Do NOT put `restart:` on a line AFTER you have listed packages because that will erase the packages that you think have been included.

Example 1: Draw three faces of the rectangular box defined by $[0, 2] \times [1, 3] \times [1, 2] = \{(x, y, z) : \ 0 \le x \le 2, \ 1 \le y \le 3, \ 1 \le z \le 2\}$ and include coordinate axes.

Let's start with the x-axis. The command 'spacecurve' has the parametric form of a curve as its argument, along with the range and choice of color. All we need here is $[t, 0, 0]$ to generate a portion of the axis. We must suppress the output with a ':' so that all the pieces may be displayed at one time. In the face labelled 'A' the y value is 1, and x and z may vary. Analyze the lines below and predict the output of each.

```
> with(plots):
> xaxis:=spacecurve([t,0,0],t=0..3,color=black):
> yaxis:=spacecurve([0,t,0],t=0..3,color=black):
> zaxis:=spacecurve([0,0,t],t=0..3,color=black):
> A:=plot3d([x,1,z],x=0..2,z=1..2,color=red):
> B:=plot3d([x,y,1],x=0..2,y=1..3,color=green):
> C:=plot3d([2,y,z],y=1..3,z=1..2,color=magenta):
> display(xaxis,yaxis,zaxis,A,B,C);
```



Example 2: Draw the portion of the paraboloid $z = 4 - x^2 - y^2$ that is over the quarter-disk of radius 2 in the first quadrant.

```
> with(plots):
> plot3d(4-x^2-y^2,x=0..2,y=0..sqrt(4-x^2),color=blue);
```



Note how we kept x between 0 and 2, but y was between 0 and $\sqrt{4-x^2}$. At this point you should have a three-cornered sheet of blue lines appear. Move the pointer onto the figure and click once. A rectangle should appear around the figure and a new set of menu options are seen above. The button $\boxed{\textbf{1:1}}$ adjusts the ratios of the axes. There are four red spheres next to $\boxed{\textbf{1:1}}$, click on each of them and note how you get different ways of showing the axes on the figure, with one option being no axes. There are 7 black spheres to the left of the four red ones. One at a time, click on each of the spheres. You may wish to end this sequence with the middle sphere. Now, click on the figure and hold down the left button of the mouse. Move the mouse so as to move the pointer and note how the figure rotates. On the left end of the line above with the spheres you will find two angles displayed. As you rotate the figure the values of those angles changes and are displayed accordingly. The angle on the left is Θ, or in lower case θ, and measures rotation about the z-axis. When $\Theta = 0$ you are looking down the x-axis. The second angle is Φ, or in lower case ϕ or φ, and measures how much the z-axis has deflected.

When $\Phi = 0$ you are looking down the z-axis from above. When $\Theta = 45$ and $\Phi = 75$, you have the x-axis to your left and the y-axis to your right equally, while the z-axis is tipped forward so as to give you the usual perspective one gets when sketching in 3-d. This will all make more sense to you after you have been introduced to cylindrical and spherical coordinate systems.

Before you move on, click carefully on the lower (right-hand) corner of the rectangle and drag it towards the opposite corner until you have a small square of about two inches, release and the figure will redraw within the box. You are expected to reduce the size of your plots in the assignments so as to save paper.

We would have gotten the same result from

```
> plot3d([x,y,4-x^2-y^2],x=0..2,y=0..sqrt(4-x^2),color=blue);
```

which is the parametric form of the same surface. The value of the parametric form is that vertical surfaces are easily handled, but of course they cannot occur as $z = f(x, y)$. Our focus here is on the quadric surfaces such as paraboloids, hyperboloids, ellipsoids, spheres and cones. But in order to display them it is best to learn how to show the result of cutting these 'solids' with planes. When one variable is held constant we have simply intersected the figure with a plane that is parallel to the coordinate plane of the remaining two variables. The result is called a 'trace'.

Example 3: Draw the solid figure bounded on the sides by $y = x$, $y = 2 - x$, and $x = 0$, below by $z = 0$, and above by $z = 4 - x^2 - y^2$.

We will use this same example to introduce double integrals later, so a little effort here will be helpful. If you draw the lines $y = x$ and $y = 2 - x$ in the first quadrant and then draw the line $x = 0$, you will see that a triangle has been formed.

BEWARE! The line $x = 0$ is vertical and is **NOT** the x-axis. Note that using the x-axis as a boundary defines a different triangle. This error frequently occurs.

Please observe the result of the following plot on your screen.

```
> plot([x,2-x],x=0..1);
```



The top line should be green and the bottom should be red. If the triangle was blue then you would have a good view looking down on the solid. But the red and green lines will be edges of the vertical surfaces. The top of our solid is the paraboloid $z = 4 - x^2 - y^2$. Keep this in mind when you consider the upper bounds for z when plotting the sides. How do we know when a 'side' is vertical? The variable z will be missing from that equation. We begin with a surface in the vertical plane $y = x$ so that every y is replaced by x.

```
> plot3d([x,x,z],x=0..1,z=0..4-2*x^2, color=red);
```

You are probably a little puzzled by the result - a vertical red line. Using what you learned up above, rotate the figure to the right 30 degrees or so. Now you should see a surface. Maple orients the initial plot so that the vertical plane $y = x$ is directly towards the viewer, i.e. $\Theta = 45$. If you examine our command you will see `[x,x,z]` which means our plot lies in the plane $y = x$ since the first and second coordinates are the same. We wish to add another plane to the situation and we will draw it separately before combining our plots.

```
> plot3d([x,2-x,z],x=0..1,z=0..4-x^2-(2-x)^2,color=green);
```

Because this surface resides in the plane $y = 2 - x$, wherever y would occur we have replaced it by $2 - x$. In particular, note the upper bound for z. To display these plots jointly, we must give the plots names and suppress their outputs with colons at the ends of their command lines.

```
> A1:=plot3d([x,x,z],x=0..1,z=0..4-2*x^2, color=red):
> A2:=plot3d([x,2-x,z],x=0..1,z=0..4-x^2-(2-x)^2,color=green):
> display(A1,A2);
```





After rotating and including the axes you should see the figure on the left above. We add our top, suppressing its output, and display all three together. When drawing the top, observe that for any fixed x, y will vary between the values x and $2 - x$, i.e. "curve-to-curve". The output is above on the right.

```
> A3:=plot3d([x,y,4-x^2-y^2],x=0..1,y=x..2-x,color=blue):
> display(A1,A2,A3);
```

We see two sides and a top of the paraboloid. The third vertical side, if it were needed, would result from the plot labelled "A4". This side is viewed by rotating to the left.

```
> A4:=plot3d([0,y,z],y=0..2,z=0..4-y^2,color=magenta):
> display(A1,A2,A3,A4);
```



Example 4: Display the portion of the hyperboloid of one sheet $x^2+y^2-z^2=1$ for which $x \le 0$, $y \ge 0$ and $-1 \le z \le 1$.

Looking at this solid from a point out on the x-axis we would see a flat surface bounded on the left by a vertical line, (the z-axis), on the top and bottom by horizontal lines (edges of planes) $z=-1$, $z=1$, and on the right by a hyperbola that bends to the left in the middle. We will draw this surface third. Rotate this surface about the z-axis $90°$ and you have a surface that will be hidden from our view, but it serves as the domain of our parametrization of the curved surface. We realize that we cannot draw the curved surface as a function of x and y. So, let's solve for y in terms of x and z. We are being careful to go 'curve-to-curve and point-to-point' here so x must vary from 0 to x as a function of z.

$$x^2+y^2-z^2=1 \quad \Rightarrow \quad y^2=1+z^2-x^2 \quad \Rightarrow \quad y=\sqrt{1+z^2-x^2}$$

Let $y=0$ and solve for x in terms of z: $x^2=1+z^2 \quad \Rightarrow \quad x=-\sqrt{1+z^2}$. In Maple, we have

```
> H1:=plot3d([x,sqrt(1+z^2-x^2),z],x=-sqrt(1+z^2)..0,z=-1..1,color=blue):
```

The top is a quarter of a disk. The radius is determined by putting $z=\pm 1$ in the equation of the hyperboloid. We get $x^2+y^2-1=1 \quad \Rightarrow \quad x^2+y^2=2$ and

```
> H2:=plot3d([x,y,1],x=-sqrt(2)..0,y=0..sqrt(2-x^2),color=red):
```

The remaining surface we need lives in the plane $x=0$. As in the domain of `H1`, z cannot be the dependent variable. The display is shown on the left below.

```
> H3:=plot3d([0,y,z], y=0..sqrt(1+z^2),z=-1..1,color=green):
> display(H1,H2,H3);
```





Example 5: Display the hyperboloid of two sheets $x^2 - y^2 - z^2 = 1$ between the planes $x = \sqrt{5}$ and $x = -\sqrt{5}$. The result is on the right above.

We list the answer without much comment. The use of 1.999 where it is clearly meant to be 2 is to avoid losing a portion of the graph when the square root of a negative number (which is meant to be 0) occurs.

```
> H5:=plot3d([-sqrt(1+y^2+z^2),y,z],y=-(1.999)..(1.999),z=-sqrt(4-y^2)..sqrt(4-y^2),
color=blue):
> H6:=plot3d([sqrt(1+y^2+z^2,y,z],y=-(1.999)..(1.999),z=-sqrt(4-y^2)..sqrt(4-y^2),
color=blue):
> H7:=plot3d([sqrt(5),y,z),y=-(1.999)..(1.999),z=-sqrt(4-y^2)..sqrt(4-y^2),color=red):
> display(H5,H6,H7);
```

C3M1 Problems

1. Use Maple to plot the coordinate axes and the remaining three faces of the box in Example 1. Rotate the output so that a portion of each face may be seen.

Use Maple to plot the regions defined in problems 2, 3 and 4.

2. R is the region in the first octant between $y = x$ and $y = 2x$ for $x \leq 2$. Also, $0 \leq z \leq 3$.

3. S is the region inside $z = x^2 + y^2$ that is below $z = 5$.

4. T is the region inside the cylinder $x^2 + y^2 = 4$ that is above $z = 0$ and below $z = 9 - x^2 - y^2$.